
1. Overview

With below changes to the TTCN source code of IMS test suite ‘iwd-TTCN3-B2015-03_D15wk38’, IMS test case 15.29 can be demonstrated to run correctly with one UE (see section 5). Execution is provided as evidence.
2. Table of Contents

31. Overview

2. Table of Contents
3
3. Verification Test Summary
3
4. Corrections required for IMS test case 15.29
3
5. Execution Log File
19
References
19

3. Verification Test Summary

Test Case:
IMS test case 15.29
ATS Version:
iwd-TTCN3-B2015-03_D15wk38
System Simulator used:
R&S CMW500
UE used:
Qualcomm MSM 8992
Verification Status:
PASS
4. Corrections required for IMS test case 15.29
Change 1 –  f_HTTP_Authentication
	Function name
	f_HTTP_Authentication

	Reason for change
	1. Globals in Http component  (containing key material computed elsewhere) needs to be accessed, see Change 2
2. BSF server domain name is bsf.home1.net, see prose specification for WWW-Authenticate header sent in Annex 29.2 step 2.

3. Initialization of security key material to be used to implement GBA key derivation (TS 33.220 clause 4.3.1) is needed

4. Variable v_PasswordGBA  not needed:

a. GBA authentication is based on the standard 3GPP AKA infratructure and the HTTP Digest authentication is according RFC 3310. See TS 33.220 clause 4.3.1. The GBA authentication in Annex 29.2 uses the same computations of nonces / responses as those used for the IMS registration.
b. The computation of the password for http digest authentication is already implemented in function fl_HTTP_CheckDigestResponse
5. Variable v_ProductTokenList used for the template for the 200 OK sent by the SS   at step 4 Annex 29.2 is incorrectly set to v_ProductTokenList := {"BSF-Server", "3gpp-gba-tmpi"}; 

6. At step 4 Annex 29.2, the SS needs to compute the correct response rspauth in the Autentication-Info header sent to the UE, see attached prose CR. This was missing.

7. Incorrect realm in the template for the Authorization header expected in step 3b Annex 29.1. 

8. The check of the digest response at step 3b in Annex 29.1 was incorrect. The key derivation for this digest response shall be based on TS 33.220 Annex B.3, see attached prose CR.

	Summary of change
	1. Runs on HTTP_PTC added
2. BSF server domain name set to bsf.home1.net
3. Initialization of security in new function fl_HTTP_PTC_Security_Init, see Change X.
4. Variable v_PasswordGBA removed. Computation of password for HTTP digest response check is implemented in local function fl_HTTP_CheckDigestResponse
5. Variable v_ProductTokenList used for the template for the 200 OK sent by the SS   at step 4 Annex 19.2 set to v_ProductTokenList := {"BSF-Server"};
6. At step 4 Annex 29.2, the SS  computes the correct response rspauth considering the XML message body needed for qualtiy of protection qop=auth-int, according to RFC2617 clause 3.2.2.3, see also attached prose CR.
7. Realm in the template for the Authorization header expected in step 3b Annex 29.1 set to "3GPP-bootstrapping@" & p_HomeDomainName, see prose spec.
8. The check of the digest response at step 3b in Annex 29.1 based on TS 33.220 Annex B.3. is implemented  via new local function fl_HTTP_CheckDigestResponse_TS33220B3.

	TTCN module
	HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

   function f_HTTP_Authentication(HTTP_DATA_PORT p_HttpDataPort,

                                 timer p_Watchdog,

                                 HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                 charstring p_PrivateUserId,

                                 charstring p_HomeDomainName,

                                 charstring p_XcapUsername) return template (value) HTTP_AuthenticationResult_Type
  { /* @sic R5-151795: restructuring of authentication procedure sic@ */

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND;

    var HTTP_DATA_IND v_InitialRequestXCAP;

    var HTTP_DATA_IND v_InitialRequestGBA;

    var HTTP_DATA_IND v_SubsequentRequestXCAP;

    var HTTP_DATA_IND v_SubsequentRequestGBA;

    var charstring v_RequestLineUri;

    var charstring v_UsernameXCAP := p_XcapUsername;

    var charstring v_UsernameGBA;

    var charstring v_PasswordXCAP;

    var charstring v_PasswordGBA;

    var Common_AuthenticationParams_Type v_AuthParams := f_IMS_AuthenticationInit(tsc_IMS_AuthRAND);

    var charstring v_Nonce := f_Bitstring2Base64(v_AuthParams.RandValue & v_AuthParams.AUTN);

    var charstring v_NextNonce := v_Nonce;

    var charstring v_CNonce;

    var ServerVal_List v_ProductTokenList;

    var template (value) WwwAuthenticate v_WwwAuthenticate;

    var template (value) AuthenticationInfo v_AuthenticationInfo;

    var template (value) ContentType v_ContentType;

    var charstring v_MessageBody;

    var template UserAgent v_UserAgent;

    var template Authorization v_Authorization;

    var charstring v_BsfServerDomainName := "bsf.operator.com";                                          // !!!! FFS: to be replaced by PIXIT ?? !!!!
    var charstring v_BTid := f_Bitstring2Base64(v_AuthParams.RandValue) & "@" & v_BsfServerDomainName;   // see 33.220 cl. 4.5.2

    var boolean v_GbaInitialRequest := true;

    var charstring v_Error := "";

    // 29.1 step 2

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := *;

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequestXCAP(v_UserAgent, v_Authorization));  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_InitialRequestXCAP := valueof(v_HTTP_DATA_IND);

    // 29.1 step 3a

    v_ProductTokenList := {"XCAP-Server"};

    v_WwwAuthenticate := fl_WwwAuthenticate_XCAP(p_AuthenticationMethod, p_HomeDomainName, v_Nonce);

    f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestXCAP.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

    if (p_AuthenticationMethod != httpDigestAndGbaAuthentication) {   // HTTP digest authentication only

      v_PasswordXCAP := "xcap";    /* @sic R5s130794 Change 25 sic@

                                      @sic R5-151795: evaluation of the password moved from f_IMS_HTTP_Transaction (IMS_XCAP_PTC) to here sic@ */

    }

    else {   // GBA authentication

      if (pc_TMPI_forGBA) {

        v_UsernameGBA := fl_GBA_TMPI_Derivation(v_AuthParams, p_PrivateUserId, v_BsfServerDomainName);

      } else {

        v_UsernameGBA := p_PrivateUserId;

      }

      // 29.2 step 1

      v_ProductTokenList := {"3gpp-gba-tmpi"};

      v_UserAgent := f_UserAgentRX(v_ProductTokenList);

      v_Authorization := fl_Authorization_GBA(v_UsernameGBA, v_BsfServerDomainName, v_GbaInitialRequest);

      v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequestXCAP(v_UserAgent, v_Authorization));  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

      if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

      v_InitialRequestGBA := valueof(v_HTTP_DATA_IND);

      // 29.2 step 2

      v_ProductTokenList := {"BSF-Server"};

      v_WwwAuthenticate := fl_WwwAuthenticate_GBA(v_Nonce);

      f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestGBA.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

      // 29.2 step 3

      v_RequestLineUri := v_InitialRequestGBA.httpRequest.requestLine.uri;    // same URI as in previous request */

      v_UserAgent := *;

      v_Authorization := fl_Authorization_GBA(v_UsernameGBA, v_BsfServerDomainName);

      v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequest(cr_HttpRequestLineXCAP(v_RequestLineUri), v_UserAgent, v_Authorization));  /* @sic R5-153795: support of ProSe - cr_HttpRequestLineXCAP sic@ */

      if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

      v_SubsequentRequestGBA := valueof(v_HTTP_DATA_IND);

      // Check authentication response

      v_PasswordGBA := "???";                                   // !!!! FFS !!!! prose does not say much about the password for GBA authentication

      v_Error := fl_HTTP_CheckDigestResponse("BSF-Server", v_SubsequentRequestGBA.httpRequest, v_PasswordGBA);
      if (v_Error != "") {

        return cs_HTTP_AuthenticationResult_Error(v_Error);

      }

      v_CNonce := f_SIP_CommaParamList_GetParamValue(v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse, "cnonce");

      // 29.2 step 4

      v_ProductTokenList := {"BSF-Server", "3gpp-gba-tmpi"};
      v_AuthenticationInfo := fl_AuthenticationInfo_GBA(v_CNonce, v_NextNonce);

      v_ContentType := cs_ContentType("application/vnd.3gpp.bsf+xml");

      v_MessageBody := oct2char(bit2oct(encvalue(cs_BootstrappingInfo(v_BTid))));

      f_HTTP_SendResponse(p_HttpDataPort,

                          v_SubsequentRequestGBA.routingInfo,

                          cs_HttpResponse_200(v_ProductTokenList, -, v_AuthenticationInfo, v_ContentType, v_MessageBody));
      v_PasswordXCAP := fl_GBA_Passwd_Derivation();

      v_UsernameXCAP := v_BTid;

    }

    // 29.1 step 3b

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := cr_AuthorizationCommon(v_UsernameXCAP, p_HomeDomainName, v_Nonce, v_InitialRequestXCAP.httpRequest.requestLine.uri, tsc_HTTP_MD5);
    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequestXCAP(v_UserAgent, v_Authorization));  /* @sic R5-153795: support of ProSe - cr_HttpRequest -> cr_HttpRequestXCAP sic@ */

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_SubsequentRequestXCAP := valueof(v_HTTP_DATA_IND);

    // Check authentication response

    v_Error := fl_HTTP_CheckDigestResponse("XCAP-Server", v_SubsequentRequestXCAP.httpRequest, v_PasswordXCAP);
    if (v_Error != "") {

      return cs_HTTP_AuthenticationResult_Error(v_Error);

    }

After change

  function f_HTTP_Authentication(HTTP_DATA_PORT p_HttpDataPort,

                                 timer p_Watchdog,

                                 HTTP_AuthenticationMethod_Type p_AuthenticationMethod,

                                 charstring p_PrivateUserId,

                                 charstring p_HomeDomainName,

                                 charstring p_XcapUsername) runs on HTTP_PTC return template (value) HTTP_AuthenticationResult_Type

  { /* @sic R5-151795: restructuring of authentication procedure sic@ */

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND;

    var HTTP_DATA_IND v_InitialRequestXCAP;

    var HTTP_DATA_IND v_InitialRequestGBA;

    var HTTP_DATA_IND v_SubsequentRequestXCAP;

    var HTTP_DATA_IND v_SubsequentRequestGBA;

    var charstring v_RequestLineUri;

    var charstring v_UsernameXCAP := p_XcapUsername;

    var charstring v_UsernameGBA;

    var charstring v_PasswordXCAP;

    // var charstring v_PasswordGBA;

    var Common_AuthenticationParams_Type v_AuthParams := f_IMS_AuthenticationInit(tsc_IMS_AuthRAND);

    var charstring v_Nonce := f_Bitstring2Base64(v_AuthParams.RandValue & v_AuthParams.AUTN);

    var charstring v_NextNonce := v_Nonce;

    var charstring v_CNonce;

    var ServerVal_List v_ProductTokenList;

    var template (value) WwwAuthenticate v_WwwAuthenticate;

    var template (value) AuthenticationInfo v_AuthenticationInfo;

    var template (value) ContentType v_ContentType;

    var charstring v_MessageBody;

    var template UserAgent v_UserAgent;

    var template Authorization v_Authorization;

    var charstring v_BsfServerDomainName :=  "bsf.home1.net"; 
    var charstring v_BTid := f_Bitstring2Base64(v_AuthParams.RandValue) & "@" & v_BsfServerDomainName;   // see 33.220 cl. 4.5.2

    var boolean v_GbaInitialRequest := true;

    var charstring v_Error := "";

    var charstring v_RspAuth;
    // #ob BZ65838 (GBA)

    fl_HTTP_PTC_Security_Init (v_AuthParams);
    log("29.1 step 2"); // @siclog 29.1 step 2 siclog@ 

    v_UserAgent := fl_UserAgentRX_XCAP();

    v_Authorization := *;

    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequest(-, v_UserAgent, v_Authorization));

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_InitialRequestXCAP := valueof(v_HTTP_DATA_IND);

    log("29.1 step 3a"); // @siclog 29.1 step 3a siclog@ 

    v_ProductTokenList := {"XCAP-Server"};

    if (p_AuthenticationMethod != httpDigestAndGbaAuthentication) {   // HTTP digest authentication only

      v_WwwAuthenticate := fl_WwwAuthenticate_XCAP(p_AuthenticationMethod, p_HomeDomainName, v_Nonce); // 

      f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestXCAP.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

      v_PasswordXCAP := "xcap";    /* @sic R5s130794 Change 25 sic@

                                      @sic R5-151795: evaluation of the password moved from f_IMS_HTTP_Transaction (IMS_XCAP_PTC) to here sic@ */

    }

    else {   // GBA authentication

      v_WwwAuthenticate := fl_WwwAuthenticate_XCAP(p_AuthenticationMethod, p_HomeDomainName, v_Nonce); // p_HomeDomainName v_BsfServerDomainName

      f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestXCAP.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

      if (pc_TMPI_forGBA) {

        v_UsernameGBA := fl_GBA_TMPI_Derivation(v_AuthParams, p_PrivateUserId, v_BsfServerDomainName);

      } else {

        v_UsernameGBA := p_PrivateUserId;

      }

      log("29.2 step 1"); // @siclog  29.2 step 1 siclog@ 

      v_ProductTokenList := {"3gpp-gba-tmpi"};

      v_UserAgent := f_UserAgentRX(v_ProductTokenList);

      v_Authorization := fl_Authorization_GBA(v_UsernameGBA, v_BsfServerDomainName, v_GbaInitialRequest);

      v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequest(-, v_UserAgent, v_Authorization));

      if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

      v_InitialRequestGBA := valueof(v_HTTP_DATA_IND);

      /*

      Authentication is based on standard 3GPP AKA infratructure which sets the password (XRES)

       used in the HTTP Digest authentication according RFC 3310.

      See TS 33.220 clause 4.3.1.

      */
      log("29.2 step 2"); // @siclog  29.2 step 2 siclog@ 

      v_ProductTokenList := {"BSF-Server"};

      v_WwwAuthenticate := fl_WwwAuthenticate_GBA(v_Nonce ); 
      f_HTTP_SendResponse(p_HttpDataPort, v_InitialRequestGBA.routingInfo, cs_HttpResponse_401(v_ProductTokenList, v_WwwAuthenticate));

      log("29.2 step 3"); // @siclog 29.2 step 3 siclog@

      v_RequestLineUri := v_InitialRequestGBA.httpRequest.requestLine.uri;    // same URI as in previous request */

      v_UserAgent := *;

      v_Authorization := fl_Authorization_GBA(v_UsernameGBA, v_BsfServerDomainName);

      v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequest(cr_HttpRequestLine(v_RequestLineUri), v_UserAgent, v_Authorization));

      if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

      v_SubsequentRequestGBA := valueof(v_HTTP_DATA_IND);

      // Check authentication response

      // v_PasswordGBA := f_IMS_PTC_Security_GetXRES();    

      v_Error := fl_HTTP_CheckDigestResponse("BSF-Server", v_SubsequentRequestGBA.httpRequest);
      if (v_Error != "") {

         return cs_HTTP_AuthenticationResult_Error(v_Error);                    

      } else {

              log("29.2 step 3 CheckDigestResponse PASS ");

      }

      v_CNonce := f_SIP_CommaParamList_GetParamValue(v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse, "cnonce");

      log("29.2 step 4"); // @siclog 29.2 step 4 siclog@

      // #ob BZ65838 (GBA) 

      v_ProductTokenList := {"BSF-Server"};
      v_MessageBody := oct2char(bit2oct(encvalue(cs_BootstrappingInfo(v_BTid))));
      v_RspAuth := f_IMS_CalculateDigestResponse("", v_SubsequentRequestGBA.httpRequest.authorization.body[0].digestResponse,

                        vc_HTTP_XRES, vc_HTTP_XRESLength, char2oct(v_MessageBody));

      v_AuthenticationInfo := fl_AuthenticationInfo_GBA(

                                 v_RspAuth, 

                                 v_CNonce, v_NextNonce, );                  

      v_ContentType := cs_ContentType("application/vnd.3gpp.bsf+xml");

      f_HTTP_SendResponse(p_HttpDataPort,

                          v_SubsequentRequestGBA.routingInfo,

                          cs_HttpResponse_200(v_ProductTokenList, -, v_AuthenticationInfo, v_ContentType, v_MessageBody));

      v_PasswordXCAP := fl_GBA_Passwd_Derivation();

      v_UsernameXCAP := v_BTid;

    }

    // @siclog 29.1 step 3b siclog@

    v_UserAgent := fl_UserAgentRX_XCAP(); 

    v_Authorization := cr_AuthorizationCommon(v_UsernameXCAP, "3GPP-bootstrapping@" & p_HomeDomainName, 

       v_Nonce, v_InitialRequestXCAP.httpRequest.requestLine.uri, tsc_HTTP_MD5);
    v_HTTP_DATA_IND := f_HTTP_ReceiveRequest(p_HttpDataPort, p_Watchdog, -, cr_HttpRequest(-, v_UserAgent, v_Authorization));

    if (not ispresent(v_HTTP_DATA_IND)) { return cs_HTTP_AuthenticationResult_Timeout; }

    v_SubsequentRequestXCAP := valueof(v_HTTP_DATA_IND);

    log("29.1 step 3b");

    // Check authentication response

    // #ob BZ65838 (GBA)

    v_Error := fl_HTTP_CheckDigestResponse_TS33220B3(v_SubsequentRequestXCAP.httpRequest, 

      v_AuthParams.CK, v_AuthParams.IK, p_HomeDomainName, tsc_IMS_AuthRAND, p_PrivateUserId);
      if (v_Error != "") {

               return cs_HTTP_AuthenticationResult_Error(v_Error);

      } else {

              log(" 29.1 step 3b CheckDigestResponse PASS ");

      }

    return cs_HTTP_AuthenticationResult_InitialRequest(v_SubsequentRequestXCAP);

  }

}
Change 2 –  new local function fl_HTTP_PTC_Security_Init
	Function name
	   fl_HTTP_PTC_Security_Init

	Reason for change
	 For GBA authentication Annex 29.2 the same mechanism  for nonce/reponses is used as for IMS registration, see attached prose CR, i.e  same XRES and XRESLength need to be computed. 

	Summary of change
	  A new function similar to the one used for IMS is defined which stores Nonce, XRES and XRESLength in component global variables

	TTCN module
	 HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

After change

    function fl_HTTP_PTC_Security_Init(template (omit) Common_AuthenticationParams_Type p_AuthenticationParams := omit

                                   ) runs on HTTP_PTC

  { 

    var Common_AuthenticationParams_Type v_AuthenticationParams;

    if (isvalue(p_AuthenticationParams)) {

      v_AuthenticationParams := valueof(p_AuthenticationParams);

      vc_HTTP_Nonce := f_Bitstring2Base64(v_AuthenticationParams.RandValue & v_AuthenticationParams.AUTN);

      vc_HTTP_XRES := v_AuthenticationParams.XRES;

      vc_HTTP_XRESLength := 4; // In octets

    } else {

                    FatalError(__FILE__, __LINE__, "fl_HTTP_PTC_Security_Init"); 

    }

  }
Change 3 –  New globals in component HTTP_PTC
	Component name
	   HTTP_PTC

	Reason for change
	  XRES and XRESLength, see change 2 above need to be globally stored.


	Summary of change
	New globals added.

	TTCN module
	 HTT_Component.ttcn

	MCC160 Comment
	


Before change

  type component HTTP_PTC { /* @status    APPROVED (IMS) */

    // port to IMS components

    port XCAP_IMS_CO_ORD_PORT IMS;

    port HTTP_CTRL_PORT HTTPCTRL;

    port HTTP_DATA_PORT HTTPDATA;

    port XCAP_PORT XCAP;     /* Simulates XCAP Server in the NW. See TS 24.623 clause 5.1 */

  };
After change

  type component HTTP_PTC { /* @status    APPROVED (IMS) */

    var charstring                         vc_HTTP_Nonce; 

    var  B32_128_Type                       vc_HTTP_XRES;             

    var integer                            vc_HTTP_XRESLength;           

    // port to IMS components

    port XCAP_IMS_CO_ORD_PORT IMS;

    port HTTP_CTRL_PORT HTTPCTRL;

    port HTTP_DATA_PORT HTTPDATA;

    port XCAP_PORT XCAP;     /* Simulates XCAP Server in the NW. See TS 24.623 clause 5.1 */

  };
Change 4 –  New fl_HTTP_CheckDigestResponse_TS33220B3
	Function name
	fl_HTTP_CheckDigestResponse_TS33220B3

	Reason for change
	  The check of the digest response at step 3b in Annex 29.1 is based on TS 33.220 Annex B.3, see Change 1 issue 8 above.

	Summary of change
	  Key derivation is implemented according  TS 33.220 Annex B.3, which is used for computing the password to be used by standard check of the digest response. 

	TTCN module
	 HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

After change

function fl_HTTP_CheckDigestResponse_TS33220B3(

                                       HttpRequest_Type p_HttpRequest,

                                       B128_Type p_CK,

                                       B128_Type p_IK,

                                       charstring p_DomainName,

                                       B128_Type v_AuthRAND,

                                       charstring p_PrivateUserId

                                      ) runs on HTTP_PTC return charstring

  {

   // #ob BZ65838 (GBA)

    var Authorization v_Authorization := p_HttpRequest.authorization;

    var charstring v_Error := "";

    // Variables for key computation

    var octetstring v_UaSecProtocolId;

    var octetstring v_NAF_Id;

    var octetstring v_S;

    var bitstring v_Ks_ext_NAF;

    if (not ischosen(v_Authorization.body[0].digestResponse)) {

      v_Error := " - No digestResponse sent in the authorization header";

      return v_Error;

    }

    // TS 32.220 Annex B.3. - NAF specific key derivation in GBA and GBA_U

    v_UaSecProtocolId := '01'O & '000000'O & '02'O; //'01'O stands for 3GPP and '02'O for the HTTP Digest from TS 24.109

    v_NAF_Id := char2oct(p_DomainName) & v_UaSecProtocolId;

    v_S := 

      '01'O &                                      // FC

      char2oct("gba-me") &                        // "gba-me"

      int2oct(lengthof(char2oct("gba-me")), 2) &     

      bit2oct(v_AuthRAND) &                        // RAND

      int2oct(lengthof(bit2oct(v_AuthRAND)), 2) &     

      char2oct(p_PrivateUserId) &                  // IMPI

      int2oct(lengthof(char2oct(p_PrivateUserId)), 2) & 

      v_NAF_Id &                                   // NAF-ID

      int2oct(lengthof(v_NAF_Id), 2); 

    v_Ks_ext_NAF :=  fx_KeyDerivationFunction ( tsc_KDF_HMAC_SHA_256,  p_CK & p_IK, v_S );  

    //Converting to correct format for password in RFC2617

    v_Ks_ext_NAF := oct2bit(char2oct(f_Bitstring2Base64(v_Ks_ext_NAF)));

    if (not f_IMS_Authorization_CheckDigestResponse(p_HttpRequest.requestLine.method, v_Authorization.body[0].digestResponse, 

                    v_Ks_ext_NAF, (lengthof(v_Ks_ext_NAF) / 8)

                    )) {

      v_Error := " DigestResponse TS 33.220 Annex B.3 check failed";

    }

    return v_Error;

  }
Change 5 –  fl_HTTP_CheckDigestResponse
	Function name
	fl_HTTP_CheckDigestResponse

	Reason for change
	  The digest response check in Annex 29.2 is the same check used for IMS registration, see attached prose CR.

	Summary of change
	  The check uses the XRES as password which has been previously computed and stored in global variables. 

	TTCN module
	 HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

 function fl_HTTP_CheckDigestResponse(charstring p_ServerName,

                                       HttpRequest_Type p_HttpRequest,

                                       charstring p_Password) return charstring

  {

    var Authorization v_Authorization := p_HttpRequest.authorization;

    var charstring v_Error := "";

    var bitstring v_XRES := oct2bit(char2oct(p_Password)); // @sic R5-151795: evaluation of the password moved from f_IMS_HTTP_Transaction (IMS_XCAP_PTC) to here sic@ */
    var integer v_XRESLength := lengthof(p_Password);

    // Check digest response

    if (not ischosen(v_Authorization.body[0].digestResponse)) {

      v_Error := p_ServerName & " - No digestResponse sent in the authorization header";

    }

    else if (not f_IMS_Authorization_CheckDigestResponse(p_HttpRequest.requestLine.method, v_Authorization.body[0].digestResponse, v_XRES, v_XRESLength)) {
      v_Error := p_ServerName & " - DigestResponse check failed";

    }

    return v_Error;

  }
After change

  function fl_HTTP_CheckDigestResponse(charstring p_ServerName,

                                       HttpRequest_Type p_HttpRequest

                                      ) runs on HTTP_PTC return charstring
  {

   /
    var Authorization v_Authorization := p_HttpRequest.authorization;

    var charstring v_Error := "";

    //var bitstring v_XRES := oct2bit(char2oct(p_Password)); // @sic R5-151795: evaluation of the password moved from f_IMS_HTTP_Transaction (IMS_XCAP_PTC) to here sic@ */

    // var integer v_XRESLength := lengthof(p_Password);

    // Check digest response

    if (not ischosen(v_Authorization.body[0].digestResponse)) {

      v_Error := p_ServerName & " - No digestResponse sent in the authorization header";

    }

    else if (not f_IMS_Authorization_CheckDigestResponse(p_HttpRequest.requestLine.method, v_Authorization.body[0].digestResponse, 

                    vc_HTTP_XRES, vc_HTTP_XRESLength)) {

      v_Error := p_ServerName & " - DigestResponse check failed";

    }

    return v_Error;

  }

Change 6 –  f_IMS_Authorization_CheckDigestResponse
	Function name
	f_IMS_CalculateDigestResponse

	Reason for change
	  The digest response in the case of qop = auth-int needs to consider the message body (entity body)  according to RFC 2617 3.2.2.3. This was missing

	Summary of change
	   A2 octets in the case of qop=auth-int according to RFC 2617 3.2.2.3 implemented.

	TTCN module
	 IMS_CommonFunctions.ttcn

	MCC160 Comment
	


Before change

  function f_IMS_Authorization_CheckDigestResponse(charstring p_Method := "REGISTER",

                                                   CommaParam_List p_DigestResponse,

                                                   bitstring p_XRES,

                                                   integer p_XRESLength) 






        return boolean

  { /* input is the comma separated parameter list: digestResponse from the Authorization header and the value of XRES calculated in ts_AuthenticationInit

       This function calculates the digest response (v_RequestDigest) for p_XRES according to RFC 2617 and compares it with the response received from the UE (v_Dresponse) */

    /* @sic R5s130794 change 6: additional parameter p_Method to allow function to be used for HTTP Digest XCAP authentication sic@ */

    var octetstring v_METHOD    := char2oct(p_Method);

    var octetstring v_Username  := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "username"));

    var octetstring v_Realm     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "realm"));

    var octetstring v_DigestUri := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "uri"));

    var octetstring v_Nonce     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nonce"));

    var octetstring v_NC        := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nc"));

    var octetstring v_CNonce    := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "cnonce"));

    var octetstring v_Qop       := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "qop"));

    var octetstring v_Passwd    := substr(bit2oct(p_XRES), 0, p_XRESLength); // @sic R5s120907 sic@

    var octetstring v_Dresponse := str2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "response"));

    var octetstring v_Secret;

    var octetstring v_HA2;

    var octetstring v_Data;

    var octetstring v_RequestDigest;

    var octetstring v_A1;

    var octetstring v_A2;

    var octetstring v_SecretData;

    var octetstring v_HEntityBody;

    // RFC 2617 3.2.2.2

    // A1 = unq(username-value) ":" unq(realm-value) ":" passwd

    v_A1 := v_Username & tsc_COLON_HEX & v_Realm & tsc_COLON_HEX & v_Passwd;

    // H(A1)

    v_Secret := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A1))));

      // A2 = Method ":" digest-uri-value

      v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri;

    v_HA2 := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A2))));

    // RFC 2617 3.2.2.1

    //   unq(nonce-value)

    //   ":" nc-value

    //   ":" unq(cnonce-value)

    //   ":" unq(qop-value)

    //   ":" H(A2)

    v_Data := v_Nonce & tsc_COLON_HEX & v_NC & tsc_COLON_HEX & v_CNonce & tsc_COLON_HEX & v_Qop & tsc_COLON_HEX & v_HA2;

    //now apply KD function described in RFC 2617 section 3.2.1

    //KD(secret, data) = H(concat(secret, ":", data))

    v_SecretData := v_Secret & tsc_COLON_HEX & v_Data;

    v_RequestDigest := fx_MD5_Hex(v_SecretData);

    return (v_RequestDigest == v_Dresponse);

  }

After change

  function f_IMS_Authorization_CheckDigestResponse(charstring p_Method := "REGISTER",

                                                   CommaParam_List p_DigestResponse,

                                                   bitstring p_XRES,

                                                   integer p_XRESLength,

                                                   octetstring p_EntityBody := ''O ) 






        return boolean

  { /* input is the comma separated parameter list: digestResponse from the Authorization header and the value of XRES calculated in ts_AuthenticationInit

       This function calculates the digest response (v_RequestDigest) for p_XRES according to RFC 2617 and compares it with the response received from the UE (v_Dresponse) */

    /* @sic R5s130794 change 6: additional parameter p_Method to allow function to be used for HTTP Digest XCAP authentication sic@ */

    var octetstring v_METHOD    := char2oct(p_Method);

    var octetstring v_Username  := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "username"));

    var octetstring v_Realm     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "realm"));

    var octetstring v_DigestUri := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "uri"));

    var octetstring v_Nonce     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nonce"));

    var octetstring v_NC        := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nc"));

    var octetstring v_CNonce    := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "cnonce"));

    var octetstring v_Qop       := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "qop"));

    var octetstring v_Passwd    := substr(bit2oct(p_XRES), 0, p_XRESLength); // @sic R5s120907 sic@

    var octetstring v_Dresponse := str2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "response"));

    var octetstring v_Secret;

    var octetstring v_HA2;

    var octetstring v_Data;

    var octetstring v_RequestDigest;

    var octetstring v_A1;

    var octetstring v_A2;

    var octetstring v_SecretData;

    var octetstring v_HEntityBody;

    // RFC 2617 3.2.2.2

    // A1 = unq(username-value) ":" unq(realm-value) ":" passwd

    v_A1 := v_Username & tsc_COLON_HEX & v_Realm & tsc_COLON_HEX & v_Passwd;

    // H(A1)

    v_Secret := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A1))));

    // #ob BZ65838, BZ65864 (GBA)

    if ( f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "qop") == "auth-int") {

       v_HEntityBody := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(p_EntityBody))));

      v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri & tsc_COLON_HEX & v_HEntityBody;

    } else {    

      // RFC 2617 3.2.2.3

      // A2 = Method ":" digest-uri-value

      v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri;

    }
    v_HA2 := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A2))));

    // RFC 2617 3.2.2.1

    //   unq(nonce-value)

    //   ":" nc-value

    //   ":" unq(cnonce-value)

    //   ":" unq(qop-value)

    //   ":" H(A2)

    v_Data := v_Nonce & tsc_COLON_HEX & v_NC & tsc_COLON_HEX & v_CNonce & tsc_COLON_HEX & v_Qop & tsc_COLON_HEX & v_HA2;

    //now apply KD function described in RFC 2617 section 3.2.1

    //KD(secret, data) = H(concat(secret, ":", data))

    v_SecretData := v_Secret & tsc_COLON_HEX & v_Data;

    v_RequestDigest := fx_MD5_Hex(v_SecretData);

    return (v_RequestDigest == v_Dresponse);

  }

Change 7 –  New function f_IMS_CalculateDigestResponse
	Function name
	f_IMS_CalculateDigestResponse

	Reason for change
	  At step 4 in Annex 29.2  we need to compute the rspauth in the Authorization Header we send to the UE. This function was missing

	Summary of change
	  Function implemented by slighlty adapting main function f_IMS_Authorization_CheckDigestResponse in modul IMS_CommonFunctions.ttcn. 

	TTCN module
	 IMS_CommonFunctions.ttcn

	MCC160 Comment
	


Before change

After change

   function f_IMS_CalculateDigestResponse(

                                      charstring p_Method := "REGISTER",

                                      CommaParam_List p_Digest,

                                       bitstring p_XRES,

                                       integer p_XRESLength,

                                       octetstring p_EntityBody := ''O

                                      ) return charstring

  {

    var octetstring v_METHOD    := char2oct(p_Method);

    var octetstring v_Username;

    var octetstring v_Realm ;

    var octetstring v_DigestUri;

    var octetstring v_Nonce;

    var octetstring v_NC;

    var octetstring v_CNonce;

    var octetstring v_Qop;

    var octetstring v_Passwd; 

    var octetstring v_Secret;

    var octetstring v_HA2;

    var octetstring v_Data;

    var octetstring v_RequestDigest;

    var octetstring v_A1;

    var octetstring v_A2;

    var octetstring v_SecretData;

    var octetstring v_HEntityBody;        

    // FFS adapted from f_IMS_Authorization_CheckDigestResponse

    v_Username  := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "username"));

    v_Realm     := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "realm"));

    v_DigestUri := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "uri"));

    v_Nonce     := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "nonce"));

    v_NC        := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "nc"));

    v_CNonce    := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "cnonce"));

    v_Qop       := char2oct(f_SIP_CommaParamList_GetParamValue (p_Digest, "qop"));

    v_Passwd    := substr(bit2oct(p_XRES), 0, p_XRESLength); 

    // RFC 2617 3.2.2.2

    // A1 = unq(username-value) ":" unq(realm-value) ":" passwd

    v_A1 := v_Username & tsc_COLON_HEX & v_Realm & tsc_COLON_HEX & v_Passwd;

    // H(A1)

    v_Secret := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A1))));

    if ( f_SIP_CommaParamList_GetParamValue (p_Digest, "qop") == "auth-int") {

       v_HEntityBody := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(p_EntityBody))));

      // RFC 2617 3.2.2.3

      //  A2 = Method ":" digest-uri-value ":" H(entity-body)

      v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri & tsc_COLON_HEX & v_HEntityBody;

    } else {    

      // RFC 2617 3.2.2.3

      // A2 = Method ":" digest-uri-value

      v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri;

    }    

    v_HA2 := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A2))));

    // RFC 2617 3.2.2.1

    v_Data := v_Nonce & tsc_COLON_HEX & v_NC & tsc_COLON_HEX & v_CNonce & tsc_COLON_HEX & v_Qop & tsc_COLON_HEX & v_HA2;

    //now apply KD function described in RFC 2617 section 3.2.1

    //KD(secret, data) = H(concat(secret, ":", data))

    v_SecretData := v_Secret & tsc_COLON_HEX & v_Data;

    v_RequestDigest := fx_MD5_Hex(v_SecretData);

    return f_StringToLower(oct2str(v_RequestDigest));

  }
Change 8 –  fl_Authorization_GBA
	Function name
	fl_Authorization_GBA

	Reason for change
	  fl_Authorization_GBA computes the Authorization header for the expected  HTTP request in Annex 29.2 step 1. Following parameters in the DigestResponse list were incorrect:
credentials=Digest -  There is no such parameter  in the core spec
digest-uri  - There is no such parameter, its name is uri

nc – string value “1” is incorrect

	Summary of change
	credentials=Digest – removed
digest-uri – replaced wth uri

nc – string value set to “00000001”, same syntax as in the IMS registration

	TTCN module
	 HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

  function fl_Authorization_GBA(charstring p_Username,

                                charstring p_BsfAddress,

                                boolean p_GbaInitialRequest := false) return template (present) Authorization

  {

    var template (present) CommaParam_List v_DigestResponse;

    var template (present) charstring v_DigestUri := ("/", "http://" & p_BsfAddress);

    v_DigestResponse := {

      cr_GenericParam("credentials",     "Digest"),
      cr_GenericParam("username",        p_Username),

      cr_GenericParam("digest-uri",      v_DigestUri)
    };

    // add further parameters for step 3

    if (p_GbaInitialRequest) {

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("nonce", ""));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("response", ""));

    }

    else {

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("realm",     tsc_GBA_Realm));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("opaque",    tsc_GBA_Opaque));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam_Common("cnonce", ?));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("nc",        "1"));
      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("algorithm", tsc_AKAv1_MD5));

    }

    return cr_Authorization_GBA(superset(all from v_DigestResponse));

  }
After change

  function fl_Authorization_GBA(charstring p_Username,

                                charstring p_BsfAddress,

                                boolean p_GbaInitialRequest := false) return template (present) Authorization

  {

    var template (present) CommaParam_List v_DigestResponse;

    var template (present) charstring v_DigestUri := ("/", "http://" & p_BsfAddress);

    v_DigestResponse := {

      // cr_GenericParam("credentials",     "Digest"),

      cr_GenericParam("username",        p_Username),

      cr_GenericParam("uri",      v_DigestUri)

    };

    // add further parameters for step 3

    if (p_GbaInitialRequest) {

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("nonce", ""));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("response", ""));

    }

    else {

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("realm",     tsc_GBA_Realm));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("opaque",    tsc_GBA_Opaque));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam_Common("cnonce", ?));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("nc",        "00000001"));

      f_CommaParam_List_Add_RX(v_DigestResponse, cr_GenericParam("algorithm", tsc_AKAv1_MD5));

    }

    return cr_Authorization_GBA(superset(all from v_DigestResponse));

  }
Change 9 –  fl_AuthenticationInfo_GBA
	Function name
	fl_AuthenticationInfo_GBA

	Reason for change
	 fl_AuthenticationInfo_GBA computes the Authentication-Info header for  HTTP response sent to the UE in Annex 29.2 step 4. Following parameters  were incorrect:

  auth-int – shall not be quotes, same syntax as in the IMS registration 

  rspauth – value for this essential token was missing, see attached prose CR  
  nc – string value “1” is incorrect
  nextnonce shall be removed, see attached prose CR

	Summary of change
	auth-int – not quoted
rspauth – value is passed as a parameter
nc – string value set to “00000001”, same syntax as in the IMS registration  and is not quoted, expected value is nc=00000001 and not nc=”00000001”
nextnonce is removed

	TTCN module
	 HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

 function fl_AuthenticationInfo_GBA(
                              charstring p_CNonce,

                               charstring p_NextNonce) return template (value) AuthenticationInfo

  {

    var template (value) CommaParam_List v_AuthInfoParams;

    v_AuthInfoParams := {

      cs_GenericParamQuoted("qop",       "auth-int"),

      cs_GenericParamQuoted("rspauth",   p_CNonce),

      cs_GenericParamQuoted("cnonce",    p_CNonce),

      cs_GenericParamQuoted("nc",        "1"),
      cs_GenericParamQuoted("opaque",    tsc_GBA_Opaque),

      cs_GenericParamQuoted("nextnonce", p_NextNonce)
    };

    return cs_AuthenticationInfo(v_AuthInfoParams);
After change

function fl_AuthenticationInfo_GBA(
                                   charstring p_RspAuth,
                                    charstring p_CNonce
                                    ) return template (value) AuthenticationInfo

  {

    var template (value) CommaParam_List v_AuthInfoParams;

    v_AuthInfoParams := {

      cs_GenericParamToken ("qop",       "auth-int"),

      cs_GenericParamQuoted("rspauth",   p_RspAuth),

      cs_GenericParamQuoted("cnonce",    p_CNonce),

      cs_GenericParamToken ("nc",         "00000001"),
      cs_GenericParamQuoted("opaque",    tsc_GBA_Opaque)
      //// cs_GenericParamQuoted("nextnonce", p_NextNonce)

    };

    return cs_AuthenticationInfo(v_AuthInfoParams);
Change 10 –  f_HTTP_ReceiveRequest
	Function name
	f_HTTP_ReceiveRequest

	Reason for change
	 According to the new attached prose CR, HTTP Requests shall match the expected sequence and not be ignored in case they not match the expected message.

	Summary of change
	Ignore code is removed, unexpected messages result in a Fail/Inconc

	TTCN module
	 HTTP_Functions.ttcn

	MCC160 Comment
	


Before change

 function f_HTTP_ReceiveRequest(HTTP_DATA_PORT p_Port,

                                 timer p_Watchdog,

                                 float p_InactivityDuration := 9999.0,

                                 template (present) HttpRequest_Type p_HttpRequest := cr_HttpRequest) return template (omit) HTTP_DATA_IND

  {

    var HTTP_DATA_IND v_DataInd;

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND := omit;

    timer t_InactivityTimer;

    t_InactivityTimer.start(p_InactivityDuration);

    alt {

      [] p_Port.receive(cr_HttpDataInd(?, p_HttpRequest)) -> value v_DataInd

        {

          v_HTTP_DATA_IND := v_DataInd;

        }

      [] p_Port.receive(cr_HttpDataInd(?, ?))

        {

              repeat;     // ignore any unexpected messages
        }

      [] t_InactivityTimer.timeout {}

      [] p_Watchdog.timeout {}

    }

    t_InactivityTimer.stop;

    return v_HTTP_DATA_IND;

  }

After change

function f_HTTP_ReceiveRequest(HTTP_DATA_PORT p_Port,

                                 timer p_Watchdog,

                                 float p_InactivityDuration := 9999.0,

                                 template (present) HttpRequest_Type p_HttpRequest := cr_HttpRequest) return template (omit) HTTP_DATA_IND

  {

    var HTTP_DATA_IND v_DataInd;

    var template (omit) HTTP_DATA_IND v_HTTP_DATA_IND := omit;

    timer t_InactivityTimer;

    t_InactivityTimer.start(p_InactivityDuration);

    alt {

      [] p_Port.receive(cr_HttpDataInd(?, p_HttpRequest)) -> value v_DataInd

        {

          v_HTTP_DATA_IND := v_DataInd;

        }

      [] p_Port.receive(cr_HttpDataInd(?, ?))

        {

              FatalError(__FILE__, __LINE__, "Http Req failed");  

        }

      [] t_InactivityTimer.timeout {}

      [] p_Watchdog.timeout {}

    }

    t_InactivityTimer.stop;

    return v_HTTP_DATA_IND;

  }

5. Execution Log File

The Qualcomm MSM 8992 UE passed this IMS test case on R&S CMW500 Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
tc_15_29_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

	R5s150946    Supporting information for agreement of IMS test case 15.29

