
1 Overview

With below changes to the TTCN source code of IMS test suite ‘iwd-TTCN3-B2015-03_D15wk50’, IMS test case 19.3.2b can be demonstrated to run correctly with one UE (see section 5). Execution is provided as evidence.

3 Verification Test Summary

Test Case:
IMS test case 19.3.2b
ATS Version:
iwd-TTCN3-B2015-03_D15wk50
System Simulator used:
DaTang LinkTester Protocol Conformance Test System
UE used:
MediaTek MT6735
Verification Status:
PASS
4 Correction to testca 19.3.2b
4.1 Change 1 –  f_TC_19_3_2b_IMS1

	Function name
	f_TC_19_3_2b_IMS1

	Reason for change
	1. To synchronize IMS PTC and IPCAN PTC. 
2. The request-URI of the INVITE request matches with the “non-emergency” number dialled.

	Summary of change
	1. Add triggers to EUTRA PTC to indicate the start and the end of IMS registration.

2. Change the indication of call type p_Context to non-Emergency call.

	TTCN module
	IMS_IRAT_19_IMS.ttcn

	MCC160 Comment
	


Before change

function f_TC_19_3_2b_IMS1() runs on IMS_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN

     */

    //@siclog "Step 1" siclog@

    var INVITE_Request v_InviteRequest;

    var charstring v_CalleeUri := "urn:service:sos";

    var charstring v_ContactUri:="urn:service:sos.country-specific";

    f_IMS_PTC_Init(PDN_1);

    f_IMS_Registration();

    // Test Case Body

    f_IMS_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    // Dial a "non emergency" number

    f_UT_RequestIMSCall(MMI, px_IMS_CalleeUri);

    //@siclog "Steps 2-4" siclog@

    v_InviteRequest := f_TC_19_1_3_TC_19_3_2x_Common(A_2_1_A7, -, v_CalleeUri, v_ContactUri);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 2 to 4");

    f_IMS_TestBody_Set(false);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

  }

After change
  function f_TC_19_3_2b_IMS1() runs on IMS_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN

     */

    //@siclog "Step 1" siclog@

    var INVITE_Request v_InviteRequest;

    var charstring v_CalleeUri := "urn:service:sos";

    var charstring v_ContactUri:="urn:service:sos.country-specific";

    f_IMS_PTC_Init(PDN_1);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // send trigger to EUTRA PTC to indicate start of IMS registration
    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // send trigger to EUTRA PTC to indicate end of IMS registration
    // Test Case Body

    f_IMS_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    // Dial a "non emergency" number

    f_UT_RequestIMSCall(MMI, px_IMS_CalleeUri);

    //@siclog "Steps 2-4" siclog@

    v_InviteRequest := f_TC_19_1_3_TC_19_3_2x_Common( -, -, px_IMS_CalleeUri, v_ContactUri);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 2 to 4");

    f_IMS_TestBody_Set(false);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

  }    
4.2 Change 2 –  f_TC_19_3_2x
	Function name
	f_TC_19_3_2x

	Reason for change
	The Service Type in EXTENDED SERVICE REQUEST message is  ‘Mobile originating CS fallback call’ in test cases 19.3.2, 19.3.2a and 19.3.2b, but ‘Mobile originating CS fallback emergency call’ in test case 19.3.2c.

	Summary of change
	Add an input parameter of service type in EXTENDED SERVICE REQUEST message.

	TTCN module
	IMS_IRAT_19_EUTRA.ttcn

	MCC160 Comment
	


Before change

  function f_TC_19_3_2x(IRAT_CoOrd_SysInfo_Type p_SysInfo,

                                  integer p_PDN) runs on EUTRA_PTC

  {

    var GutiParameters_Type v_GutiParams; // @sic R5s140788 sic@

    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo; //@sic R5s150061 sic@

    var boolean v_AckReceived;

    var boolean v_TimerExpired;

    var boolean v_EsrReceived;

    timer t_WaitforExtServiceReq := 2.0;  // @sic R5-145792 sic@ // @sic R5-150699 sic@ @sic R5-150700 sic@ @sic R5-150704 sic@

    // EXCEPTION: Within 2 seconds of step 7 the UE may transmit EXTENDED SERVICE REQUEST

    t_WaitforExtServiceReq.start; // @sic R5-145792 sic@

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    v_AckReceived := false;

    v_TimerExpired := false;

    v_EsrReceived := false;

    // @sic R5-150699 sic@ @sic R5-150700 sic@ @sic R5-150704 sic@

    // If CS Fallback is perfomed, the UE sends Extended service request with Service type set to mobile originating CS fallback emergency call as defined in 24.301 clause 9.9.3.27

    alt { // @sic R5-145792 sic@

      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, // @sic R5s140788 sic@

                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_EXT_SERVICE_REQUEST_Common(f_EUTRA_SecurityKSIasme_Get(),

                                                                                         tsc_EST_CsFallback_EC,

                                                                                         bit2oct(v_GutiParams.M_TMSI),

                                                                                         omit))))

        {

          t_WaitforExtServiceReq.stop;

          v_EsrReceived := true;

          if (not v_AckReceived) {

            v_TimerExpired:= true;

            repeat;

          }

        }

      [] t_WaitforExtServiceReq.timeout  // CS fallback is not developed: ESR is not received.

        {

          if (not v_AckReceived) {

            v_TimerExpired := true;

            repeat;

          }

        }

      [] IMS[p_PDN].receive(cmr_IMS_IPCAN_Trigger) // //Receive Trigger indicating receipt of ACK in parallel with EXTENDED SERVICE REQUEST

        {

          v_AckReceived := true;

          if(not v_TimerExpired) {

            repeat;             // In case not ESR received and not timeout, then repeat until one of these conditions is satisfied.

          }

        }

    }

    // IF px_RATComb_Tested = EUTRA_UTRA UE performs CS fallback or cell reselection to a cell supporting CS domain (UTRAN) and performs emergency(19.1.3, 19.3.2, 19.3.2c)/normal (19.3.2b) call in CS domain together with MM/GMM registration

    if (pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_EUTRA_SendAuthParameters(eutra_CellA, UTRAN, true, true);

      if (not v_EsrReceived) { // CS fallback is not developed.

        f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

      }

      else { // CS fallback is developed

        if (f_IRAT_GetUTRAN_FDD_TDD_Mode() == UTRAN_FDD) {

          v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraFdd(p_SysInfo.Utran[0].FDD.Freq);      //[0] corresponds to Utran cell 5 according to the Init-function. //@sic R5s150061 sic@

        }

        else {

          v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraTdd(p_SysInfo.Utran[0].TDD.Freq);

        }

        f_EUTRA_RRC_ConnectionRelease_Common(eutra_CellA, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other, v_RedirectedCarrierInfo));

      }

      f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE); //@sic R5s150273 sic@

      f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    }

    //IF px_RATComb_Tested = EUTRA_GERAN UE performs CS fallback or cell reselection to a cell supporting CS domain (GERAN cell) and performs emergency (19.1.3, 19.3.2, 19.3.2c)/normal (19.3.2b) call in CS domain

    else if (pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_EUTRA_SendAuthParameters(eutra_CellA, GERAN, true, true);

      if (not v_EsrReceived) { // CS fallback is not developed.

        f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

      }

      else {

        v_RedirectedCarrierInfo := cs_RedirectionInfo_Geran_NoNeighCells(p_SysInfo.Geran[0].Arfcn, //@sic R5s150061 sic@

                                                                         p_SysInfo.Geran[0].BandIndicator );

        f_EUTRA_RRC_ConnectionRelease_Common(eutra_CellA, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other, v_RedirectedCarrierInfo));

      }

      f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE); //@sic R5s150273 sic@

      f_IRAT_WaitForCoOrd_Trigger(GERAN);

    }

    else {

      FatalError(__FILE__, __LINE__, "Invalid RAT combination configured");

    }

  }
After change 
  function f_TC_19_3_2x(IRAT_CoOrd_SysInfo_Type p_SysInfo,

                                  integer p_PDN,

                                  EPS_ServiceTypeValue p_EPS_ServiceTypeValue := tsc_EST_CsFallback_EC) runs on EUTRA_PTC //add by hanson @20151113 for cs fallback emergency

  {

    var GutiParameters_Type v_GutiParams; // @sic R5s140788 sic@

    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo; //@sic R5s150061 sic@

    var boolean v_AckReceived;

    var boolean v_TimerExpired;

    var boolean v_EsrReceived;

    timer t_WaitforExtServiceReq := 2.0;  // @sic R5-145792 sic@ // @sic R5-150699 sic@ @sic R5-150700 sic@ @sic R5-150704 sic@

    // EXCEPTION: Within 2 seconds of step 7 the UE may transmit EXTENDED SERVICE REQUEST

    t_WaitforExtServiceReq.start; // @sic R5-145792 sic@

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    v_AckReceived := false;

    v_TimerExpired := false;

    v_EsrReceived := false;

    // @sic R5-150699 sic@ @sic R5-150700 sic@ @sic R5-150704 sic@

    // If CS Fallback is perfomed, the UE sends Extended service request with Service type set to mobile originating CS fallback emergency call as defined in 24.301 clause 9.9.3.27

    alt { // @sic R5-145792 sic@

      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, // @sic R5s140788 sic@

                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_EXT_SERVICE_REQUEST_Common(f_EUTRA_SecurityKSIasme_Get(),

                                                                                         p_EPS_ServiceTypeValue,

                                                                                         bit2oct(v_GutiParams.M_TMSI),

                                                                                         omit))))

        {

          t_WaitforExtServiceReq.stop;

          v_EsrReceived := true;

          if (not v_AckReceived) {

            v_TimerExpired:= true;

            repeat;

          }

        }

      [] t_WaitforExtServiceReq.timeout  // CS fallback is not developed: ESR is not received.

        {

          if (not v_AckReceived) {

            v_TimerExpired := true;

            repeat;

          }

        }

      [] IMS[p_PDN].receive(cmr_IMS_IPCAN_Trigger) // //Receive Trigger indicating receipt of ACK in parallel with EXTENDED SERVICE REQUEST

        {

          v_AckReceived := true;

          if(not v_TimerExpired) {

            repeat;             // In case not ESR received and not timeout, then repeat until one of these conditions is satisfied.

          }

        }

    }

    // IF px_RATComb_Tested = EUTRA_UTRA UE performs CS fallback or cell reselection to a cell supporting CS domain (UTRAN) and performs emergency(19.1.3, 19.3.2, 19.3.2c)/normal (19.3.2b) call in CS domain together with MM/GMM registration

    if (pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_EUTRA_SendAuthParameters(eutra_CellA, UTRAN, true, true);

      if (not v_EsrReceived) { // CS fallback is not developed.

        f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

      }

      else { // CS fallback is developed

        if (f_IRAT_GetUTRAN_FDD_TDD_Mode() == UTRAN_FDD) {

          v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraFdd(p_SysInfo.Utran[0].FDD.Freq);      //[0] corresponds to Utran cell 5 according to the Init-function. //@sic R5s150061 sic@

        }

        else {

          v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraTdd(p_SysInfo.Utran[0].TDD.Freq);

        }

        f_EUTRA_RRC_ConnectionRelease_Common(eutra_CellA, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other, v_RedirectedCarrierInfo));

      }

      f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE); //@sic R5s150273 sic@

      f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    }

    //IF px_RATComb_Tested = EUTRA_GERAN UE performs CS fallback or cell reselection to a cell supporting CS domain (GERAN cell) and performs emergency (19.1.3, 19.3.2, 19.3.2c)/normal (19.3.2b) call in CS domain

    else if (pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_EUTRA_SendAuthParameters(eutra_CellA, GERAN, true, true);

      if (not v_EsrReceived) { // CS fallback is not developed.

        f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

      }

      else {

        v_RedirectedCarrierInfo := cs_RedirectionInfo_Geran_NoNeighCells(p_SysInfo.Geran[0].Arfcn, //@sic R5s150061 sic@

                                                                         p_SysInfo.Geran[0].BandIndicator );

        f_EUTRA_RRC_ConnectionRelease_Common(eutra_CellA, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other, v_RedirectedCarrierInfo));

      }

      f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE); //@sic R5s150273 sic@

      f_IRAT_WaitForCoOrd_Trigger(GERAN);

    }

    else {

      FatalError(__FILE__, __LINE__, "Invalid RAT combination configured");

    }

  }
4.3 Change 3 –  f_TC_19_3_2b_EUTRA

	Function name
	f_TC_19_3_2b_EUTRA

	Reason for change
	The Service Type in EXTENDED SERVICE REQUEST message is  ‘Mobile originating CS fallback call’ in test case 19.3.2b.

	Summary of change
	Set the Service Type in EXTENDED SERVICE REQUEST message to  ‘Mobile originating CS fallback call’.

	TTCN module
	IMS_IRAT_19_EUTRA.ttcn

	MCC160 Comment
	


Before change

  function f_TC_19_3_2b_EUTRA() runs on EUTRA_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN */

    var IRAT_CoOrd_SysInfo_Type v_SysInfo;  //@sic R5s150061 sic@

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c10); // @sic R5-131470 sic@ @sic R5s140788 ch.1

    if (px_RATComb_Tested == EUTRA_UTRA) {

      v_SysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);

    }

    else {

      v_SysInfo := f_IRAT_WaitForCoOrd_SysInfo (GERAN);

    }

    // Create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA, HANDLE_UM_DATA, USE_BIG_GRANTS);    /* @sic R5s130681: DISCARD_UM_DATA; R5s140160: USE_BIG_GRANTS sic@

                                                                               @sic R5s150338: f_EUTRA_CellConfig_DRB_Common replaced by f_EUTRA_CellConfig_Def sic@

                                                                               @sic R5-153746: DISCARD_UM_DATA -> HANDLE_UM_DATA sic@ */

    // power on the cell

    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    //Send trigger to let other RAT know that preamble is done so can change cell power to suitable cell @sic R5s140788 sic@

    if (px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);

    }

    else if(px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_SendCoOrd(GERAN, cms_IRAT_Trigger);

    }

    else  {

      FatalError(__FILE__, __LINE__, "Invalid RAT combination configured");

    }

    f_EUTRA_TestBody_Set(true);

    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(eutra_CellA, -, -, -, -, -, -, -, -, -, cr_EstablishmentCause_MoData);

    // Coordination betweenIMS PTC and EUTRA PT: step 4--> Note1

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]); //Wait reception of IMS ACK on step 4 // @sic R5-150704 sic@

    //@siclog "Steps 4- 5b4" siclog@

    f_TC_19_3_2x(v_SysInfo, tsc_Index_PDN1);

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_ReleaseAllCells();

  }
After change
  function f_TC_19_3_2b_EUTRA() runs on EUTRA_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN */

    var IRAT_CoOrd_SysInfo_Type v_SysInfo;  //@sic R5s150061 sic@

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c10); // @sic R5-131470 sic@ @sic R5s140788 ch.1

    if (px_RATComb_Tested == EUTRA_UTRA) {

      v_SysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);

    }

    else {

      v_SysInfo := f_IRAT_WaitForCoOrd_SysInfo (GERAN);

    }

    // Create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA, HANDLE_UM_DATA, USE_BIG_GRANTS);    /* @sic R5s130681: DISCARD_UM_DATA; R5s140160: USE_BIG_GRANTS sic@

                                                                               @sic R5s150338: f_EUTRA_CellConfig_DRB_Common replaced by f_EUTRA_CellConfig_Def sic@

                                                                               @sic R5-153746: DISCARD_UM_DATA -> HANDLE_UM_DATA sic@ */

    // power on the cell

    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    //Send trigger to let other RAT know that preamble is done so can change cell power to suitable cell @sic R5s140788 sic@

    if (px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);

    }

    else if(px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_SendCoOrd(GERAN, cms_IRAT_Trigger);

    }

    else  {

      FatalError(__FILE__, __LINE__, "Invalid RAT combination configured");

    }

    f_EUTRA_TestBody_Set(true);

    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(eutra_CellA, -, -, -, -, -, -, -, -, -, cr_EstablishmentCause_MoData);

    // Coordination betweenIMS PTC and EUTRA PT: step 4--> Note1

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]); //Wait reception of IMS ACK on step 4 // @sic R5-150704 sic@

    //@siclog "Steps 4- 5b4" siclog@

    f_TC_19_3_2x(v_SysInfo, tsc_Index_PDN1, tsc_EST_CsFallback_MO);  //normal call cs fallback

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_ReleaseAllCells();

  }
4.4 Change 4 –  f_TC_19_3_2b_GERAN

	Function name
	f_TC_19_3_2b_GERAN

	Reason for change
	E-UTRAN will send trigger to let other RAT know that preamble is done so can change cell power to suitable cell.

	Summary of change
	Wait for the trigger sent from E-UTRAN.

	TTCN module
	IMS_IRAT_19_GERAN.ttcn

	MCC160 Comment
	


Before change

  function f_TC_19_3_2b_GERAN() runs on GERAN_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN */

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_Send_IR_SysInfoToEUTRA (geran_Cell24); // @sic R5s150061 sic@

    f_GERAN_CreateCell(geran_Cell24);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell

    f_GERAN_CreateTCH (geran_Cell24);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_ServingCell);

    //@siclog "Step 5b1 - 5b3" siclog@

    f_GERAN_CSFB_Redirection_CCO_M0 (geran_Cell24); // new in 19_3_2b

    f_GERAN_TestBody_Set(false);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);     // Tell EUTRA test is over

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F);

  }
After change
  function f_TC_19_3_2b_GERAN() runs on GERAN_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN */

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_Send_IR_SysInfoToEUTRA (geran_Cell24); // @sic R5s150061 sic@

    f_GERAN_CreateCell(geran_Cell24);

f_IRAT_WaitForCoOrd_Trigger(EUTRA); //Wait for the UE to be attached to EUTRAN
    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell

    f_GERAN_CreateTCH (geran_Cell24);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    f_GERAN_SetChPwrLevel (geran_Cell24, tsc_TchCh, tsc_GERAN_ServingCell);

    //@siclog "Step 5b1 - 5b3" siclog@

    f_GERAN_CSFB_Redirection_CCO_M0 (geran_Cell24); // new in 19_3_2b

    f_GERAN_TestBody_Set(false);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);     // Tell EUTRA test is over

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G1_IDLE, -, -, tsc_TCH_F);

  }
4.5 Change 5 –  f_TC_19_3_2b_UTRAN

	Function name
	f_TC_19_3_2b_UTRAN

	Reason for change
	E-UTRAN will send trigger to let other RAT know that preamble is done so can change cell power to suitable cell.

	Summary of change
	Wait for the trigger sent from EUTRAN.

	TTCN module
	IMS_IRAT_19_UTRAN.ttcn

	MCC160 Comment
	


Before change

    function f_TC_19_3_2b_UTRAN() runs on UTRAN_IRAT_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN */

    //Initialise test case

    f_UTRAN_Init ( EUTRA_UTRAN );

    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5); // @sic R5s150061 sic@

    //Create UTRAN cell5

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationSuitableNeighbourCell);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    f_UTRAN_TestBody_Set( true );

    //@siclog "Step 9" siclog@

    f_UTRAN_CS_Fallback_WithRedirection(utran_Cell5, false); // new in 19_3_2b

    //@siclog "Step 10" siclog@

    //The SS releases the UE connection

    f_UTRAN_CC_Disconnect (utran_Cell5);

    f_UTRAN_RRC_ConnRel ( utran_Cell5, cell_Dch );

    f_UTRAN_RRC_ConnEst(utran_Cell5);

    f_UTRAN_GMM_RAU(utran_Cell5, false);

    f_UTRAN_RRC_ConnRel( utran_Cell5, cell_Dch );

    f_UTRAN_TestBody_Set ( false );

    // Switch off/power off and release cells

    // f_UTRAN_IP_Handling_Change_Configuration(utran_CellDedicated, cell_DCH_64kPS_RAB_SRB); // @sic R5s140788, R5s150736 now done in postamble sic@

    f_UTRAN_Postamble(utran_Cell5, U1_IDLE);

    // send coordination message for end of test case

    f_IRAT_SendCoOrd ( EUTRA, cms_IRAT_Trigger );

  } // f_TC_19_3_2b_UTRAN
After change
 function f_TC_19_3_2b_UTRAN() runs on UTRAN_IRAT_PTC

  { /* Non-UE detectable emergency call / IM CN sends a 380 with unavailable emergency service URN / UE performs normal call via CS domain / UTRAN or GERAN */

    //Initialise test case

    f_UTRAN_Init ( EUTRA_UTRAN );

    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5); // @sic R5s150061 sic@

    //Create UTRAN cell5

f_UTRAN_SS_CreateCellDCH(utran_Cell5);
//Send the default System Information messages

f_UTRAN_SendDefSysInfo(utran_Cell5);
f_IRAT_WaitForCoOrd_Trigger(EUTRA); //Wait for the UE to be attached to EUTRAN
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationSuitableNeighbourCell);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    f_UTRAN_TestBody_Set( true );

    //@siclog "Step 9" siclog@

    f_UTRAN_CS_Fallback_WithRedirection(utran_Cell5, false); // new in 19_3_2b

    //@siclog "Step 10" siclog@

    //The SS releases the UE connection

    f_UTRAN_CC_Disconnect (utran_Cell5);

    f_UTRAN_RRC_ConnRel ( utran_Cell5, cell_Dch );

    f_UTRAN_RRC_ConnEst(utran_Cell5);

    f_UTRAN_GMM_RAU(utran_Cell5, false);

    f_UTRAN_RRC_ConnRel( utran_Cell5, cell_Dch );

    f_UTRAN_TestBody_Set ( false );

    // Switch off/power off and release cells

    // f_UTRAN_IP_Handling_Change_Configuration(utran_CellDedicated, cell_DCH_64kPS_RAB_SRB); // @sic R5s140788, R5s150736 now done in postamble sic@

    f_UTRAN_Postamble(utran_Cell5, U1_IDLE);

    // send coordination message for end of test case

    f_IRAT_SendCoOrd ( EUTRA, cms_IRAT_Trigger );

  } // f_TC_19_3_2b_UTRAN
4.6 Change 6 –  f_GetNetworkFeatureSupport

	Function name
	f_GetNetworkFeatureSupport

	Reason for change
	In test case 19.3.2x, emergency bearer services in S1 mode are not supported.

	Summary of change
	Set the emcBS  of network support feature to 1.

	TTCN module
	EPS_NAS_Templates.ttcn

	MCC160 Comment
	


Before change

  function f_GetNetworkFeatureSupport () return template (value) EPS_NetworkFeatureSupport

  {

    var template (value) EPS_NetworkFeatureSupport v_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice;

    // @sic R5s130681 sic@

    select (f_GetTestcaseAttrib_Eutra_Release(testcasename())) { // @sic R5s130508 sic@

      case ("","rel_10") { // rel8 or rel-10 - 36.508 just says rel-9

      }

      case ("rel_9") {

        v_NtwkFeatSupport.imsVoPS := '1'B; // @sic R5s130902 sic@

        v_NtwkFeatSupport.emcBS := '1'B; // @sic R5s130977 sic@

      }

      case else {}

    }

    return v_NtwkFeatSupport;

  }
After change
  function f_GetNetworkFeatureSupport () return template (value) EPS_NetworkFeatureSupport
  {

    var template (value) EPS_NetworkFeatureSupport v_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice;

    // @sic R5s130681 sic@

    select (f_GetTestcaseAttrib_Eutra_Release(testcasename())) { // @sic R5s130508 sic@

      case ("","rel_10") { // rel8 or rel-10 - 36.508 just says rel-9

      }

      case ("rel_9") {

        v_NtwkFeatSupport.imsVoPS := '1'B; // @sic R5s130902 sic@

        v_NtwkFeatSupport.emcBS := '1'B; // @sic R5s130977 sic@

      }

      case else {}

    }


//NetworkFeatureSupport 19.3.2x, should not support emergency bearer services in S1 mode


select (testcasename()) {

  case ("TC_19_3_2") { v_NtwkFeatSupport.emcBS := '0'B; }


  case ("TC_19_3_2a") { v_NtwkFeatSupport.emcBS := '0'B; }


  case ("TC_19_3_2b") { v_NtwkFeatSupport.emcBS := '0'B; }


  case ("TC_19_3_2c") { v_NtwkFeatSupport.emcBS := '0'B; }


  case else{}


}
    return v_NtwkFeatSupport;

  }
5 Execution Log File

The MediaTek MT6735 UE passed this IMS test case on DaTang LinkTester Protocol Conformance Test System. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_19_3_2b_E40_G900.html
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

